
1. Overview

This document lists all the essencial changes needed to correct problems in the TTCN implementation of test case 22.4.19a which is part of the NBIOT test suite in the ‘iwd-TTCN3-B2018-03_D18wk37’ ATS delivery. 
With these changes applied the test case can be demonstrated to run with one or more UEs (see section 4). Execution log files are provided as evidence.

3. Verification Test Summary

Test Case:
22.4.19a
Test Group:
\NBIOT\22_4\NBIOT_RRC_ControlPlane.ttcn
ATS Version:
iwd-TTCN3-B2018-03_D18wk37
System Simulator used:
Rohde & Schwarz CMW500




UE used:
MediaTek MT2625
Verification Status:
PASS


4. Corrections required for NBIOT TC 22.4.19a
Change 1 f_TC_22_4_19a
	Function name
	f_TC_22_4_19a

	Reason for change
	1. The setting up of the RACH procedure list assumes that the UE would send the RRC Connection Re-establishment Request only once at Step 2. However, during verification it was found that the UE re-sends the RRC Connection Re-establishment Request multiple times. The UE initiates a RACH procedure each time when it wishes the sends the RRC Connection Re-establishment Request. However, the SS tries to send the RRC Connection Setup message as per the RACH procedure list for the second RRC Connection Re-establishment Request message. This is incorrect.
2. As AS security is not enabled, it is not necessary to release RRC security. A reset of the SRB1bis is needed on NCell1 to prepare for the TAU at the end of the test. Also, it is recommended to perform this before power for NCell1 is switched off. It is also essential to ensure that the ACK for the ATTACH COMPLETE is sent to the UE before the SRB1bis is reset otherwise the UE would consider the ATTACH procedure as failed and would try to re-initaite ATTACH at Step 4-8 instead of responding to Paging.
3. As mentioned in point 1 above, the UE initiates multiple RRC Connection Re-establishment Requests, the TTCN should have code to handle these additional requests.

4. AS security is not enabled in this TC, hence it does not make sense to activate RRC security on NCell2 after Step 3.

	Summary of change
	1. Removed the implementation of the RACH procedure list and replaced it with the explicit calls to function f_NBIOT_SS_ConfigRachProcedureMsg4 to configure no response at Step 3 and to configure RRC Connection Setup at Step 4-8
2. Removed the call to function f_NBIOT_RRC_ConnectionRelease_Local and replaced it with function f_NBIOT_SS_SRBs_DRBs_Reset for NCell1 and allowed for a delay of 500ms to ensure the ACK for ATTACH COMPLETE is sent.
3. Added TTCN code to handle multiple RRC Connection Re-establishment Requests

4. Removed the code for activating RRC security on NCell2

	TTCN module
	NBIOT\22_4\NBIOT_RRC_ControlPlane.ttcn

	MCC160 Comment
	


Before Change:

	function f_TC_22_4_19a_NBIOT() runs on NBIOT_PTC

  {// NB-IoT / Radio link failure / T301 expiry / T311 expiry / RRC connection re-establishment

    var SystemInformationBlockType2_NB_r13 v_SIB2;

    var NBIOT_SecurityParams_Type v_SecurityParams;

    var template (value) NB_RachProcedureList_Type v_RachProcedureList;

    var template (value) NB_CellPowerList_Type v_CellPowerList_Step1 := {

      cs_NB_CellPower(nbiot_Cell1, tsc_NonSuitableOffNBIOTCellRS_EPRE),

      cs_NB_CellPower(nbiot_Cell2, tsc_SuitableCellRS_EPRE)

    };

    var template (value) NB_CellPowerList_Type v_CellPowerList_Step9 := {

      cs_NB_CellPower(nbiot_Cell1, tsc_NonSuitableOffNBIOTCellRS_EPRE),

      cs_NB_CellPower(nbiot_Cell2, tsc_NonSuitableOffNBIOTCellRS_EPRE)

    };

    var template (value) NB_CellPowerList_Type v_CellPowerList_Step11 := {

      cs_NB_CellPower(nbiot_Cell1, tsc_ServingCellRS_EPRE),

      cs_NB_CellPower(nbiot_Cell2, tsc_NonSuitableOffNBIOTCellRS_EPRE)

    };

    var float v_T301 := f_NBIOT_SetTimerToleranceMin(rrcTimer, 2.5);

    f_NBIOT_Init(c2);

    //Set the value of T301 and T311 timer in SIB2 for Ncell 1, Ncell 2

    v_SIB2 := f_NBIOT_CellInfo_GetSIB2(nbiot_Cell1);

    v_SIB2.ue_TimersAndConstants_r13.t301_r13 := ms2500;

    v_SIB2.ue_TimersAndConstants_r13.t311_r13 := ms10000;

    v_SIB2.cp_Reestablishment_r14 := true_;

    f_NBIOT_CellInfo_SetSIB2 (nbiot_Cell1, v_SIB2);

    v_SIB2 := f_NBIOT_CellInfo_GetSIB2(nbiot_Cell2);

    v_SIB2.ue_TimersAndConstants_r13.t301_r13 := ms2500;

    v_SIB2.ue_TimersAndConstants_r13.t311_r13 := ms10000;

    v_SIB2.cp_Reestablishment_r14 := true_;

    f_NBIOT_CellInfo_SetSIB2 (nbiot_Cell2, v_SIB2);

    //Create and configure all cells

    f_NBIOT_CellConfig_Def(nbiot_Cell1, CONTROL_PLANE);

    f_NBIOT_CellConfig_Def(nbiot_Cell2, CONTROL_PLANE);

    //Bring UE to initial state

    f_NBIOT_Preamble(nbiot_Cell1, CONTROL_PLANE);

    f_NBIOT_TestBody_Set(true);

    //Setup the no response in step 3 and the RRCConnectionSetup for the TAU procedure in steps 4-8

    v_RachProcedureList := {

      cs_NB_RachProcedure(cs_NB_RandomAccessResponseConfig_Def, cs_NB_ContentionResolutionCtrl_Def(cs_ContentionResolution_DlschSdu_None)),

      cs_NB_RachProcedure(cs_NB_RandomAccessResponseConfig_Def, cs_NB_ContentionResolutionCtrl_Def(cs_ContentionResolution_DlschSdu(f_RRC_DL_CCCH_NB_Encvalue(f_NBIOT_508_RRCConnectionSetup()))))

    };

    f_NBIOT_SS_CommonCellConfig(nbiot_Cell2, cads_NB_RachProcedure_Config_REQ(nbiot_Cell2, cs_TimingInfo_Now, -, -, v_RachProcedureList));

    //@siclog "Step 1" siclog@

    //The SS changes Ncell 1 to "Off" and Ncell2 to Suitable

    f_NBIOT_SetCellPowerList(v_CellPowerList_Step1);

    //Release RRC security, SRBs and DRBs. Reset the SRBs and DRBs ready for TAU at end of test

    f_NBIOT_RRC_ConnectionRelease_Local(nbiot_Cell1, CONTROL_PLANE, f_NBIOT_GetNextSearchSpace(nbiot_Cell1));

    //@siclog "Step 2" siclog@

    //The UE sends an RRC Connection Reestablishment Request-NB message on Ncell2

    f_NBIOT_RRC_RRCConnectionReestablishmentRequest_r14(nbiot_Cell1, nbiot_Cell2); // @sic R5-183199 sic@

    //@siclog "Step 3" siclog@

    //The SS does not respond for 2.5s (T301)

    f_Delay(v_T301);

    //Activate RRC security

    v_SecurityParams := f_NBIOT_Security_Get();

    v_SecurityParams := f_NBIOT_SS_AS_ReestablishResumeSecurity(nbiot_Cell2, v_SecurityParams, rrcConnectionReestablishment);

    f_NBIOT_Security_Set(v_SecurityParams);

    //@siclog "Step 4-8" siclog@

    //Does the test result of generic test procedure in TS 36.508 subclause 8.1.5A.2 indicate that the UE is camped on E-UTRAN Ncell 2?

    f_NBIOT_508Check_CP_ResponseToPagingForMTAccess(nbiot_Cell2);// @sic R5-183199 sic@

    f_NBIOT_PreliminaryPass(__FILE__, __LINE__, "Steps 4 - 8");

    f_Delay (0.5); // to ensure Service Accept is sent

    //@siclog "Step 9" siclog@

    //The SS changes Ncell 2 to "Off"

    f_NBIOT_SetCellPowerList(v_CellPowerList_Step9);

    //@siclog "Step 10" siclog@

    //Wait for 12 seconds (T311 = 10s)

    f_Delay(12.0);

    //@siclog "Step 11" siclog@

    //The SS changes Ncell 1 to ServingCell

    f_NBIOT_SetCellPowerList(v_CellPowerList_Step11);

    //@siclog "Step 12-16" siclog@

    //Check: Does the test result of generic test procedure in TS 36.508 subclause 8.1.5A.2 indicate that the UE is camped on E-UTRAN Ncell 1?

    //NOTE: The UE performs a TAU procedure and the RRC connection is released.

    f_NBIOT_508Check_CP_ResponseToPagingForMTAccess(nbiot_Cell1);// @sic R5-183199 sic@

    //@siclog "Step 17" siclog@

    //@sic R5-183199 sic@

    //The SS transmits an RRCConnectionRelease-NB message.

    f_NBIOT_RRC_ConnectionRelease(nbiot_Cell1);

    f_NBIOT_TestBody_Set(false);

    //Switch/power off UE

    f_NBIOT_Postamble(nbiot_Cell1, CONTROL_PLANE, N1_IDLE);

  }




After Change:

	function f_TC_22_4_19a_NBIOT() runs on NBIOT_PTC

  {// NB-IoT / Radio link failure / T301 expiry / T311 expiry / RRC connection re-establishment

    var SystemInformationBlockType2_NB_r13 v_SIB2;

    var NBIOT_SecurityParams_Type v_SecurityParams;

    var template (value) NB_RachProcedureList_Type v_RachProcedureList;

    var template (value) NB_CellPowerList_Type v_CellPowerList_Step1 := {

      cs_NB_CellPower(nbiot_Cell1, tsc_NonSuitableOffNBIOTCellRS_EPRE),

      cs_NB_CellPower(nbiot_Cell2, tsc_SuitableCellRS_EPRE)

    };

    var template (value) NB_CellPowerList_Type v_CellPowerList_Step9 := {

      cs_NB_CellPower(nbiot_Cell1, tsc_NonSuitableOffNBIOTCellRS_EPRE),

      cs_NB_CellPower(nbiot_Cell2, tsc_NonSuitableOffNBIOTCellRS_EPRE)

    };

    var template (value) NB_CellPowerList_Type v_CellPowerList_Step11 := {

      cs_NB_CellPower(nbiot_Cell1, tsc_ServingCellRS_EPRE),

      cs_NB_CellPower(nbiot_Cell2, tsc_NonSuitableOffNBIOTCellRS_EPRE)

    };

    timer t_Wait;
    f_NBIOT_Init(c2);

    //Set the value of T301 and T311 timer in SIB2 for Ncell 1, Ncell 2

    v_SIB2 := f_NBIOT_CellInfo_GetSIB2(nbiot_Cell1);

    v_SIB2.ue_TimersAndConstants_r13.t301_r13 := ms2500;

    v_SIB2.ue_TimersAndConstants_r13.t311_r13 := ms10000;

    v_SIB2.cp_Reestablishment_r14 := true_;

    f_NBIOT_CellInfo_SetSIB2 (nbiot_Cell1, v_SIB2);

    v_SIB2 := f_NBIOT_CellInfo_GetSIB2(nbiot_Cell2);

    v_SIB2.ue_TimersAndConstants_r13.t301_r13 := ms2500;

    v_SIB2.ue_TimersAndConstants_r13.t311_r13 := ms10000;

    v_SIB2.cp_Reestablishment_r14 := true_;

    f_NBIOT_CellInfo_SetSIB2 (nbiot_Cell2, v_SIB2);

    //Create and configure all cells

    f_NBIOT_CellConfig_Def(nbiot_Cell1, CONTROL_PLANE);

    f_NBIOT_CellConfig_Def(nbiot_Cell2, CONTROL_PLANE);

    //Bring UE to initial state

    f_NBIOT_Preamble(nbiot_Cell1, CONTROL_PLANE);

    f_NBIOT_TestBody_Set(true);

    //Setup the no response in step 3 and the RRCConnectionSetup for the TAU procedure in steps 4-8

    f_NBIOT_SS_ConfigRachProcedureMsg4(nbiot_Cell2); //WA#22_4_19a
    //Reset the SRBs and DRBs ready for TAU at end of test
    f_Delay (0.5); // to ensure ACK for the ATTACH COMPLETE is sent WA#22_4_19a
    f_NBIOT_SS_SRBs_DRBs_Reset(nbiot_Cell1, CONTROL_PLANE); //WA#22_4_19a
    //@siclog "Step 1" siclog@

    //The SS changes Ncell 1 to "Off" and Ncell2 to Suitable

    f_NBIOT_SetCellPowerList(v_CellPowerList_Step1);

    //@siclog "Step 2" siclog@

    //The UE sends an RRC Connection Reestablishment Request-NB message on Ncell2

    f_NBIOT_RRC_RRCConnectionReestablishmentRequest_r14(nbiot_Cell1, nbiot_Cell2); // @sic R5-183199 sic@

    f_NBIOT_SS_SRBs_DRBs_Reset(nbiot_Cell2, CONTROL_PLANE); //WA#22_4_19a
    //@siclog "Step 3" siclog@

    //The SS does not respond for 2.5s (T301)

    //WA#22_4_19a

    t_Wait.start(f_NBIOT_SetTimerToleranceMin(rrcTimer, 2.5));

    alt { 

      [] SRB.receive(car_NB_SRB0_RrcPdu_IND(nbiot_Cell2, cr_RRCConnectionReestablishmentRequest_NB_r14(?, otherFailure)))

        {

        
f_NBIOT_SS_SRBs_DRBs_Reset(nbiot_Cell2, CONTROL_PLANE); 

        
repeat;

        }

      [] t_Wait.timeout {}

    }
    f_NBIOT_SS_ConfigRachProcedureMsg4(nbiot_Cell2, f_NBIOT_508_RRCConnectionSetup()); //WA#22_4_19a
    //@siclog "Step 4-8" siclog@

    //Does the test result of generic test procedure in TS 36.508 subclause 8.1.5A.2 indicate that the UE is camped on E-UTRAN Ncell 2?

    f_NBIOT_508Check_CP_ResponseToPagingForMTAccess(nbiot_Cell2);// @sic R5-183199 sic@

    f_NBIOT_PreliminaryPass(__FILE__, __LINE__, "Steps 4 - 8");

    f_Delay (0.5); // to ensure Service Accept is sent

    //@siclog "Step 9" siclog@

    //The SS changes Ncell 2 to "Off"

    f_NBIOT_SetCellPowerList(v_CellPowerList_Step9);

    //@siclog "Step 10" siclog@

    //Wait for 12 seconds (T311 = 10s)

    f_Delay(12.0);

    //@siclog "Step 11" siclog@

    //The SS changes Ncell 1 to ServingCell

    f_NBIOT_SetCellPowerList(v_CellPowerList_Step11);

    //@siclog "Step 12-16" siclog@

    //Check: Does the test result of generic test procedure in TS 36.508 subclause 8.1.5A.2 indicate that the UE is camped on E-UTRAN Ncell 1?

    //NOTE: The UE performs a TAU procedure and the RRC connection is released.

    f_NBIOT_508Check_CP_ResponseToPagingForMTAccess(nbiot_Cell1);// @sic R5-183199 sic@

    //@siclog "Step 17" siclog@

    //@sic R5-183199 sic@

    //The SS transmits an RRCConnectionRelease-NB message.

    f_NBIOT_RRC_ConnectionRelease(nbiot_Cell1);

    f_NBIOT_TestBody_Set(false);

    //Switch/power off UE

    f_NBIOT_Postamble(nbiot_Cell1, CONTROL_PLANE, N1_IDLE);

  }
    


Change 2 f_NBIOT_RRC_RRCConnectionReestablishmentRequest_r14
	Function name
	f_NBIOT_RRC_RRCConnectionReestablishmentRequest_r14

	Reason for change
	The default alignment for the variable v_CellIdentity which is passed into the external function fx_NasIntegrityAlgorithmBitstring is OctetAligned. The CellIdentity is 28 bits and should be passed as Unaligned 28 bits rather than an OctetAligned parameter.

	Summary of change
	The import lines are swapped to import the EUTRA_ASN1_CellIdentity_Type as “PerUnaligned”

	TTCN module
	NBIOT\22_4\NBIOT_RRC_ControlPlane.ttcn

	MCC160 Comment
	


Before Change:

	import from NBIOT_Imported_EUTRA_ASN1_Types all;

  


After Change:

	import from NBIOT_Imported_EUTRA_ASN1_Types all except { type EUTRA_ASN1_CellIdentity_Type };

import from NBIOT_Imported_EUTRA_ASN1_Types { type EUTRA_ASN1_CellIdentity_Type } with {encode "PerUnaligned"};




5. Execution Log Files

5.1 MediaTek MT2625
The MediaTek MT2625 UE passed this NBIOT TC on Rohde & Schwarz CMW500.The documentation below is enclosed as evidence of the successful test case run [1]:

Test Case Execution log file: 
Rohde-Schwarz\tc_22_4_19a_MTK2625.log 
(Note: PICS/PIXIT settings are captured at the beginning of the TLI log)

In the log file (in .txt / -html format) the complete test case execution can be seen. All TLI events are presented and message contents are fully decoded and can be verified. Preliminary verdicts and the final test case verdict can be seen in the log file

References

	[1]
	R5s180540
 This archive comprises Execution log files, PICS/PIXIT files



